```
> with(qseries):
> with(ramamocktheta):
```

THE PROC FOR COMPUTING THE COEFFICIENTS OF THE THIRD ORDER MOCK THETA FUNCTION f(q)

```
>
    comp_f3:=proc(T)
      local f3A,aA,k,n,n1,m,k1,k2,f3Aq,f3AMAT,j;
      f3A:=Array(0..T):
      aA:=Array(0..T):

  ##################################################################
  ######
      # INITIALIZE Arrays f3A and aA
      for k from 0 to T do aA[k]:=0: f3A[k]:=0: od:
      f3A[0]:=1: aA[0]:=1:

  ##################################################################
  ######
  #   We compute the coefficients a(n) and store them in the Array
  Aa
        for n from 1 to trunc( (sqrt(1+24*T)-1)/6) do
           n1:=n*(3*n+1)/2:
           for m from 0 to trunc( (T-n1)/n ) do
              k:=n1+n*m:
              aA[k]:= aA[k]+ 4*(-1)^(m+n):
            od:
        od:

  ##################################################################
  ######
      # Compute coeffs of f3 and store them in Array f3A
        for n from 1 to T do
        if modp(n,1000)=0 then print(n); fi:
          k1:=trunc ((1 + sqrt(1+24*n))/6):
          k2:= trunc( (-1 +sqrt(1+24*n))/6):
          for k from 1 to k1 do
            f3A[n]:=f3A[n] - (-1)^k*f3A[n-k*(3*k-1)/2]:
          od:
        for k from 1 to k2 do
            f3A[n]:=f3A[n] - (-1)^k*f3A[n-k*(3*k+1)/2]:
          od:
          f3A[n]:=f3A[n]+aA[n]:
        od:

  ##################################################################
  ######
      # SAVING COEFFICIENTS IN DIFFERENT FORMATS
  #   (1) Saving coefficients as a matrix in a txt file using
  ExportMatrix.
        f3AMAT:=Matrix(T+1,2):
        for j from 1 to T do
           f3AMAT[j,1]:=j-1:
           f3AMAT[j,2]:=f3A[j-1]:
        od:
        ExportMatrix("f3AMAT.txt",f3AMAT);
  #   (2) Saving the Array f3A in a dot-m file:
```

```
      save f3A,"f3A.m":
#   (3) Saving the series f3Aq in a dot-m file:
    f3Aq:=add(f3A[n]*q^n,n=0..T):
      save f3Aq, "f3funcs.m":
    RETURN():
    end:
```

This PROC is based on the following identity om p.64 of Watson's paper.

$$f(q)\prod_{r=1}^{\infty}(1-q^r)=1+4\sum_{n=1}^{\infty}\frac{(-)^n q^{\frac{1}{2}n(3n+1)}}{1+q^n},$$

$$\phi(q)\prod_{r=1}^{\infty}(1-q^r)=1+2\sum_{n=1}^{\infty}\frac{(-)^n(1+q^n)\,q^{\frac{1}{2}n(3n+1)}}{1+q^{2n}},$$

$$\chi(q)\prod_{r=1}^{\infty}(1-q^r)=1+\sum_{n=1}^{\infty}\frac{(-)^n(1+q^n)\,q^{\frac{1}{2}n(3n+1)}}{1-q^n+q^{2n}}.$$

The series expansion of $\prod_{r=1}^{\infty}(1-q^r) = 1 + \sum_{k=1}^{\infty}(-1)^k q^{\frac{k(3k-1)}{2}} + \sum_{k=1}^{\infty}(-1)^k q^{\frac{k(3k+1)}{2}}$

See RED ARROW above. In the PROC comp_f3 there are TWO arrays f3A and aA.
The array f3A gives the coefficients of the function f(q). This means that

f(q) = $\sum_{n=0}^{\infty} f3A[n]\, q^n$ except it is not really an infinite sum n only goes up to n=T

On the RHS the coefficient of q^n is aA[n]. So aA[0]=1

$$\sum_{n=1}^{\infty}\frac{(-1)^n q^{\frac{n(3n+1)}{2}}}{(1+q^n)} = \sum_{n=1}^{\infty}(-1)^n q^{\frac{n(3n+1)}{2}}\sum_{m=0}^{\infty}(-1)^m q^{mn}$$

$$= \sum_{n=1}^{\infty}\sum_{m=0}^{\infty}(-1)^{n+m} q^{\frac{n(3n+1)}{2}+mn}$$

This means aA[k] = sum of $(-1)^{n+m}$ over k satisfying k = n(3n+1)/2 + mn

FIRST we see that aA[0] =1. Then we need to calculuate aA[k] for k > 0.

We need only consider m and n where n(3*n+1)/2 + m*n <= T, where n>=1 and m>=0
To find the biggest value of n to consider we solve

```
> solve(n*(3*n+1)/2 = T, n);
```

$$-\frac{1}{6}+\frac{\sqrt{1+24\,T}}{6},\ -\frac{1}{6}-\frac{\sqrt{1+24\,T}}{6} \tag{1}$$

This explains the first loop n=1 to trunc((sqrt(1+24*T)-1)/6)
For this value of n we calculate how big m can be:
Let n1 = n*(3*n+1)/2. Then we have n1 + m*n <= T so that m<= (T-n1)/n.
This explains the loop m=0 to trunc((T-n1)/n).
Let k = n*(3*n+1)/2 + m*n = n1 + m*n and 4*(-1)^(m+n) gets added to the value of aA[k].

THIS EXPLAINS THE FULL DOUBLE LOOP:

```
for n from 1 to trunc( (sqrt(1+24*T)-1)/6) do
        n1:=n*(3*n+1)/2:
        for m from 0 to trunc( (T-n1)/n ) do
          k:=n1+n*m:
          aA[k]:= aA[k]+ 4*(-1)^(m+n):
         od:
     od:
```

NEXT WE EXPLAIN THE DOUBLE LOOP THAT CALCULATES each f3A[k].
We multiply the following

$$\prod_{r=1}^{\infty}\left(1-q^{r}\right) = 1 + \sum_{k=1}^{\infty}(-1)^{k} q^{\frac{k(3k-1)}{2}} + \sum_{k=1}^{\infty}(-1)^{k} q^{\frac{k(3k+1)}{2}}$$

by f(q) $= \sum_{n=0}^{\infty} f3A[n]\, q^{n}$

We get

$$\sum_{n=0}^{\infty} f3A[n]\, q^{n} + \sum_{m=0}^{\infty}\sum_{k=1}^{\infty} f3A[m]\cdot(-1)^{k}\, q^{m+\frac{k(3k-1)}{2}} + \sum_{m=0}^{\infty}\sum_{k=1}^{\infty} f3A[m]\cdot(-1)^{k}\, q^{m+\frac{k(3k+1)}{2}}$$

For each n we calculate the coefficient of q^n in this series
We need to know how big k is in the second sum. We solve

```
> solve(k*(3*k-1)/2 = T,k);
```

$$\frac{1}{6}+\frac{\sqrt{1+24\,T}}{6},\ \frac{1}{6}-\frac{\sqrt{1+24\,T}}{6} \tag{2}$$

This explains the loop k=1 to k1 where k1:=trunc ((1 + sqrt(1+24*n))/6)

We do the same in the third sum. We solve

```
> solve(k*(3*k+1)/2 = T,k);
```

$$-\frac{1}{6}+\frac{\sqrt{1+24\,T}}{6},\ -\frac{1}{6}-\frac{\sqrt{1+24\,T}}{6} \tag{3}$$

This explains the loop k=1 to k1 where k2:=trunc ((-1 + sqrt(1+24*n))/6)

So on the LHS the coefficient of q^n is
f3A[n] + sum f3A[n-k*(3*k-1)/2]*(-1)^k for k=1 to k1
+ sum f3A[n-k*(3*k+1)/2]*(-1)^k for k=1 to k2
This HAS TO EQUAL the coefficient of q^n on RHS which is aA[n].
Hence

f3A[n] = aA[n] - sum f3A[n-k*(3*k-1)/2]*(-1)^k for k=1 to k1 - sum f3A[n-k*(3*k+1)/2]*(-1)^k for k=1 to k2

This gives a recurrence for f3A[n] with initialization f3A[0]=1.
PUTTING ALSO THIS TOGETHER EXPLAINS THE DOUBLE LOOP

```
for n from 1 to T do
     if modp(n,1000)=0 then print(n); fi:
       k1:=trunc ((1 + sqrt(1+24*n))/6):
       k2:= trunc( (-1 +sqrt(1+24*n))/6):
       for k from 1 to k1 do
```

```
    f3A[n]:=f3A[n] - (-1)^k*f3A[n-k*(3*k-1)/2]:
  od:
for k from 1 to k2 do
    f3A[n]:=f3A[n] - (-1)^k*f3A[n-k*(3*k+1)/2]:
  od:
  f3A[n]:=f3A[n]+aA[n]:
od:
```